
18

DEZVOLTAREA DE APLICAŢII ORIENTATE OBIECT IN LIMBAJ DE ASAMBLARE

18.1. Concepte folosite în dezvoltarea programelor orientate obiect

O clasă este definită ca abstractizare  a trăsăturilor esenţiale ale unei colecţii de obiecte înrudite. Deci o clasă este o mulţime de obiecte care partajează o structură comună şi un comportament comun. Se extinde astfel conceptul datelor de tip structură, combinându-se datele cu codul ce le procesează.

O clasă are două componente, una statică şi alta dinamică. Prima componentă se referă la proprietăţile obiectelor caracterizate de clasă, şi este reprezentată de datele membre (câmpurile cu nume care posedă valoare şi care caracterizează starea obiectului în timpul execuţiei). A doua se referă la comportamentul obiectelor clasei, deci la modul în care acţionează şi reacţionează acestea, şi este constituită din metodele clasei respective.

Un obiect este definit ca materializarea sau concretizarea tipologiei descrise de o anumită clasă. El este un concept, abstractizare sau lucru cu limite precizate şi cu înţeles pentru problema în cauză.

Figura 18.1 – Relaţia dintre o clasă şi obiectele ei


Relaţia dintre o clasă şi obiectele acesteia este echivalentă relaţiei dintre o variabilă şi tipul acesteia, subliniindu-se astfel ideea că obiectele şi clasa ce le defineşte nu sunt altceva decât un model mai avansat de variabile şi o extensie a tipurilor fundamentale specifice fiecărui limbaj. Fiecare obiect preia şablonul impus de clasa de care aparţine, şablon ce descrie caracteristicile sale (componenta statică, reprezentată de date), precum şi operaţiile ce sunt permise asupra acestora (interfaţa obiectului, descrisă prin metodele clasei). Operaţia de creare a unui obiect din informaţiile exprimate de o clasă, se numeşte instanţiere, iar operaţia inversă (definirea unei clase pornind de la o mulţime de obiecte) poartă numele de clasificare sau tipizare.


Prin intermediul metodelor, orice obiect este o entitate dinamică care poate fi creată, utilizată şi distrusă. Ca metode cu semnificaţie deosebită, atrag atenţia constructorii şi destructorii. Aşa cum ne sugerează şi denumirea, constructorii sunt acele operaţii apelate pentru crearea obiectelor şi iniţializarea stării lor, în timp ce destructorii realizează operaţia inversă, ştergând spaţiul de memorie ocupat de metodele şi datele membre ale acestora.


Acest mod de organizare a datelor şi a metodelor, de manipulare a acestora, face apel la o serie de concepte precum abstractizarea, moştenirea şi polomorfismul.
Abstractizarea – reprezintă procesul de ignorare intenţionată a detaliilor nesemnificative şi reţinerea proprietăţilor definitorii ale unei entităţi. Astfel, se ascund detaliile de implementare, dezvăluindu-se, prin intermediul interfeţei obiectului, comportarea esenţială a acestuia relativ la viziunea exterioară pe care vrea să o imprime creatorul clasei.

Încapsularea – este mijlocul de separare a informaţiilor de manipulare a unei entităţi (aspectele externe, accesibile utilizatorului acesteia) de informaţiile de implementare. Ea poate fi întâlnită şi în alte tehnici de programare cum este, spre exemplu, modularizarea programelor prin folosirea bibliotecilor. Prin intermediul acesteia se realizează protejarea anumitor date membre ale obiectelor împotriva distrugerii lor accidentale, ajungându-se în acest fel la o robusteţe ridicată a programelor în implementarea orientată obiect comparativ cu cea tradiţională.

Moştenirea – se defineşte ca fiind o relaţie între clase prin intermediul căreia o clasă partajează structura şi comportamentul definite de una sau mai multe clase. În funcţie de numărul claselor de la se porneşte moştenirea, cunoscute sub numele de clase de bază sau superclase, putem avea moştenire simplă (când avem o singură clasă de bază) sau moştenire multiplă (când se folosesc caracteristicile mai multor clase). Clasa rezultantă se numeşte subclasă sau clasă derivată.


Figura 18.2 – Moştenirea multiplă

Polimorfismul – este capacitatea unei entităţi de a îmbrăca mai multe forme. În programarea clasică, funcţiile se diferenţiază prin numele sub care sunt cunoscute, parametri pe care îi acceptă, precum şi valoarea pe care o returnează.  În programarea orientată obiect compilatorul poate să facă diferenţa între funcţii şi prin intermediul clasei de care aparţin. 


Funcţii care au acelaşi nume, parametri şi valoare de retur în clasa de bază, pot fi redefinite (supraîncărcate) în clasele derivate şi ajustate la noul context în care sunt apelate. Acest tip de polimorfism poartă denumirea de polimorfism de moştenire.

În cadrul polimorfismului ad-hoc, de care aparţine cel menţionat mai sus, se regăseşte şi polimorfismul coerciziune care este caracteristic limbajelor de programare care dispun de facilităţi de conversie internă între tipuri.

Un alt tip de polimorfism este şi polimorfismul parametric, dar care apelând la capacitatea compilatoarelor tradiţionale de a face diferenţa dintre funcţii cu nume şi valoare de retur identice, dar cu tip şi / sau număr de parametri diferit, nu aparţine neapărat de programarea orientată obiect.

Alături de aceste concepte mai este adăugat şi cel de persistenţă a obiectelor. Acesta se referă la timpul de viaţă al unui obiect, la capacitatea acestuia de a permite salvarea pe un suport de memorie externă şi de reconstituire ulterioară în memoria internă a calculatorului. Realizarea acestei cerinţe presupune identificarea clasei obiectului din informaţiile salvate şi construirea dinamică (în timpul execuţiei) a acestuia, acţiune ce necesită metode avansate de programare cum ar fi RTTI (Run-Time Type Identification - identificarea tipului în timpul execuţiei) şi folosirea metodelor virtuale.

Prin apelarea la aceste concepte, programarea orientată obiect aduce o serie de avantaje:

· reutilizabilitate – se permite reutilizarea codului deja testat şi se creşte, astfel, robusteţea programelor concomitent cu diminuarea duratei de realizare a acestora;

· scăderea complexităţii – fragmentarea aplicaţiei în entităţi şi relaţii, care au înţeles pentru utilizator, este o analiză convenţională şi o tehnică de design, care duce la o diminuare semnificativă a complexităţii iniţiale a problemei studiate.

· flexibilitate – prin intermediul moştenirii se permite adăugarea sau extragerea de obiecte, fără schimbări profunde în logica aplicaţiei, realizându-se astfel o extindere sau o acomodare a aplicaţiei cu minim de efort.

· înţelegere mai uşoară – punerea în corespondenţă unu la unu a obiectelor din structura aplicaţiei cu cele ale problemei studiate duce la o înţelegere mai uşoară a fenomenelor atât de către utilizator cât şi de către proiectant.

Toate aceste avantaje au ca rezultat o creştere a productivităţii şi a uşurinţei în întreţinere a aplicaţiilor orientate obiect, ceea ce impune tehnica de elaborare a programelor orientată obiect ca soluţia optimă pentru programatori şi proiectanţi de sisteme.

Orice tip de dată elementară sau derivată şi orice mecanism de gestionare a resurselor unui sistem de calcul se regăsesc mai întâi în descriptorii respectivi, funcţii şi operaţii în limbaj de asamblare.

Dezvoltarea de aplicaţii orientate obiect face excepţie numai aparent de la această cerinţă. Proprietăţile obiectelor sunt implementate mai întâi în limbajele evoluate (cum ar fi C++, Pascal) şi corespondenţele lor în modulele obiect sunt suplinite de proprietăţile şi funcţiile primare deja existente în limbajul de asamblare.

În [SWAN95] se prezintă modalitatea de dezvoltare a aplicaţiilor orientate obiect fără ca limbajul de asamblare să conţină atribuite, structuri specifice definirii de clase şi manipulării de obiecte.

Pentru descrierea mai multor modalităţi de implementare a programării orientate obiect în limbajul de asamblare, se va face apel la reprezentarea tipurilor de date întregi şi complexe şi a metodelor de manipulare a acestora. S-a ales astfel un exemplu simplu care să uşureze înţelegerea şi urmărirea codului prezentat, tehnicile folosite fiind uşor de extins la orice aplicaţie reală.

18.2. Definirea obiectelor prin structuri

Una dintre cele mai simple metode de implementare a obiectelor în limbajul de asamblare este folosirea regulilor de construcţie a structurilor simple, fără folosirea unor construcţii specifice orientate obiect.

Se consideră o aplicaţie în care se manipulează numere întregi şi numere complexe (s-au ales numerele întregi în locul celor reale pentru o simplificare a codului). Se defineşte mai întâi o clasă care să încapsuleze comportamentul numerelor întregi şi, derivată din aceasta, una care să definească comportamentul numerelor complexe. 

Pentru fiecare din cele două clase se definesc operaţiile de adunare şi de înmulţire (cele de scădere şi de împărţire sunt asemănătoare). Pentru clasa Integer se dezvoltă următoarele proceduri:

...

.code

adunare_integer proc


push ax


mov ax, [di].nr_int


add ax, word ptr [si]


mov [di].nr_int, ax 


pop ax


retn

adunare_integer endp

inmultire_integer proc


push ax


mov ax, [di].nr_int


mul word ptr [si]


mov [di].nr_int, ax 


pop ax


retn

inmultire_integer endp

...

Numărul implicat în adunare / înmulţire este trimis prin adresă prin registrul SI, iar adresa obiectului este trimisă prin registrul DI.

Pentru clasa Complex procedurile sunt:

...

adunare_complex proc


push ax


mov ax, [di].nr_int


add ax, word ptr [si]


mov [di].nr_int, ax 


mov ax, [di].nr_complex


add ax, word ptr [si+2]


mov [di].nr_complex, ax 


pop ax


retn

adunare_complex endp

inmultire_complex proc

.data


int_temp dw 0

.code


push ax


mov ax, [di].nr_complex


mul word ptr [si+2]


mov bx, ax


mov ax, [di].nr_int


mul word ptr [si]


sub ax, bx


mov int_temp, ax


mov ax, [di].nr_complex


mul word ptr [si]


mov bx, ax


mov ax, [di].nr_int


mul word ptr [si+2]


add ax, bx



mov [di].nr_complex, ax



mov ax, int_temp


mov [di].nr_int, ax


pop ax


retn

inmultire_complex endp

...

Numărul complex este format din două date membre, nr_int şi nr_complex, adunarea şi înmulţirea fiind definite astfel:

Transmiterea parametrilor se face prin adresa din registrul SI ([SI] = partea întreagă de adunat / înmulţit, iar [SI+2] = partea imaginară), iar adresa obiectului se trimite prin registrul DI.

După definirea procedurilor urmează definirea obiectelor:

...

.data


Integer struc



nr_int 

dw
0



adun_int
dw
adunare_integer



inm_int
dw
inmultire_integer


Integer ends


Complex struc



baza

Integer <>



nr_complex
dw
0



adun_compl
dw
adunare_complex




inm_compl
dw
inmultire_complex


Complex ends

...

Obiectele sunt definite ca nişte structuri obişnuite, având ca membri atât datele cât şi pointeri spre metode. Obiectul Complex este derivat din obiectul Integer prin definirea unei date membre de tip Integer (baza Integer <>).

Pentru rularea celor două clase se poate construi un cod de genul celui următor:

...


ob_integer Integer <4>


ob_complex Complex <7,5>


numar1     dw      10


numar2
    dw
   5

.code

start:


mov ax,@data


mov ds,ax


lea  si, numar1


;[si] = numar1, [si+2] = numar2



lea  di,ob_integer



call ob_integer.adun_int
;nr_int = 4+10 = 14


call ob_integer.inm_int
;nr_int = 14*10 = 140


lea  di,ob_complex


call ob_complex.adun_compl
;nr_int = 7+10 = 17






;nr_complex = 5+5 = 10


call ob_complex.inm_compl
;nr_int = 17*10-10*5 = 120






;nr_complex = 17*5+10*10 = 185

sfarsit:


mov ax,4c00h


int 21h

end start

Instanţierea claselor se face ca în cazul structurilor obişnuite. Se observă că înaintea apelurilor procedurilor este necesară încărcarea în registrele SI şi DI a parametrilor şi a obiectelor asupra cărora se efectuează operaţia.

Pentru verificarea corectitudinii calculelor se încarcă programul executabil, rezultat prin asamblarea codului de mai sus, în turbo debugger dacă se foloseşte asamblorul oferit de firma Borland. Pentru aceasta trebuie ca programul să fie asamblat şi link-editat cu următoarele comenzi:

tasm /zi <nume_program.asm>

tlink  /v  <nume_program.obj>

18.3. Definirea obiectelor prin macrodefiniţii

Limbajul de asamblare conţine facilităţi de definire a macrodefiniţiilor. Macrodefiniţiile sunt construcţii realizate cu ajutorul unor cuvinte cheie al căror efect constă într-un proces de expandare în textul programului sursă, permiţându-se în acest mod definirea în mod simbolic a unor secvenţe de program (instrucţiuni, definiţii de date, directive etc.), asociate cu un nume. 

Macrodefiniţiile sunt structuri complexe în care se includ şi elemente ale macroexpandării condiţionale, repetitive, recursive şi stringizare (vezi lucrările [SOMN92], [MUS96]). Un aspect foarte important de care trebuie sa se ţină cont în utilizarea macroinstrucţiunilor este faptul că ele sunt procesate într-o fază preliminară asamblării.

Pentru definirea structurilor necesare modelării orientate obiect, precum şi a moştenirii şi încapsulării datelor acestora, există şi varianta folosirii macrodefiniţiilor. Macrodefiniţiile sunt deosebit de utile în realizarea programelor orientate obiect, putând fi folosite construcţii speciale de verificare a restricţiilor care să genereze eventualele erori nu în timpul execuţiei programului, ci în faza de analizare sintactică a codului sursă, exact ca în limbajele de nivel înalt orientate obiect.

În continuare se prezintă un model de implementare a obiectelor în limbajul de asamblare, ce face apel la un sistem de simboluri variabile, construite prin stringizare, folosite împreună cu elemente macrocondiţionale pentru verificarea restricţiilor impuse de modelul obiectual. Sintaxa rezultată prezintă asemănări evidente cu limbajele de nivel înalt orientate obiect, în special cu limbajul C++.

Mai întâi este necesară definirea unor date şi a unor variabile simbolice folosite în cadrul macrodefiniţiilor (semnificaţia acestora va rezulta ulterior, pe parcursul descrierilor din paginile următoare):

; V1.0 

.data

       currOffset dw 0
;offsetul obiectului apelator pt. o ;functie din aceeasi clasa

;VARIABILE DE CONSTRUIRE (GENERALE)

currProt = 0

;implicit public (folosita in declaratiile ;claselor)

currId = 0

;id-ul curent (id-ul primei clase = 1)

currClass = 0

;(nici o clasa initial)

currApelator = 0
;apelatorul curent al operatiei (id-ul ;obiectului)

Id = 1


;folosita de OBJECTS

gasit = 0

;folosita la cautarea in clase de baza

deplasament = 0
;deplasamentul folosit in cazul ;accesarii unor variabile din




;clasele de baza

baseId = 0

;id-ul clasei de baza la cautarea datelor ;membre

publicDer = 1

;indica daca derivarea se face ;public (=1) sau private (=0)

;VARIABILE CARE DEPIND DE CLASE (de numele clasei sau de id-ul ei)

;<numeClasa>Id

id-ul clasei cu numele <numeClasa>

;<numeObiect>Id

id-ul obiectului (EGAL ;INTOTDEAUNA CU ID-UL CLASEI

;



DE CARE APARTINE!)

;no<Id> 


nr elementelor din clasa cu id-ul <id>

;baseNo<Id>


nr claselor de baza pentru ;clasa cu id-ul <id>

;disable<Id>Prot

indica daca se foloseste sau nu ;protectia datelor

;



(folosita in apelul functiilor ;membre)

;friend<Id>Of<currId>Id
indica, daca exista, prietenia ;intre clasele cu

;



id-urile <Id> si <curcId>

;depl<baseId>In<currId>
indica deplasarea clasei de baza ( <baseId> ) in clasa

;



cu id-ul <currId>

Pentru definirea unei clase se foloseşte macrodefiniţia class cu următoarea sintaxă:
class <nume_clasa>

...

<nume_clasa> ends

Această macrodefiniţie încapsulează o structură standard, iniţiind şi o serie de simboluri variabile ce vor fi folosite la identificarea clasei:
class MACRO nume

IFB <nume>



%OUT < **Error** Class name not specified ... >



.ERR


ELSE
;genereaza clasa




%OUT < Creating class nume ... >




currId = currId + 1

;id-ul urmator va fi +1




nume&Id = currId

;id-ul de clasa




;creaza variabila ce contine nr de elemente


setBaseAndNoToZero   %currId





;creaza variabila de control a protectiei ;clasei




setDisableProt   %currId, 0
;initial protectia este ;activata




nume struc


;creaza structura


ENDIF

ENDM

În cazul în care nu se precizează un nume pentru clasă se generează eroare. În caz contrar acest nume va fi folosit pentru declararea structurii, nu înainte ca id-ul clasei (format prin stringizare din numele clasei şi şirul de caractere “Id”) se fie stabilit în mod secvenţial cu ajutorul variabilei simbolice <currId>. Astfel, toate clasele vor avea un număr de identificare distinct (începând cu 1).

Macrodefiniţia class se foloseşte de alte două macrodefiniţii. Una dintre aceste este setBaseAndNoToZero:

setBaseAndNoToZero MACRO id


baseNo&id = 0


No&Id = 0

ENDM

care se foloseşte de parametrul transmis (<id> = id-ul clasei) pentru a stabili numărul de clase de bază (baseNo<id>) şi numărul de elemente (No<Id>) ale clasei definite ca fiind zero.

Altă macrodefiniţie folosită este setDisableProt:

setDisableProt MACRO id, val


disable&id&Prot = val

ENDM

cu ajutorul căreia se creează o variabilă de protecţie a datelor ce va fi folosită în macrodefiniţiile următoare (prin transmiterea parametrului val cu valoarea 0 protecţia este activă).

Protecţia membrilor clasei se realizează cu ajutorul variabilei simbolice <currProt> a cărei valoare se poate modifica cu macrodefiniţiile următoare:

public_ MACRO


currProt = 0

ENDM

private_ MACRO


currProt = 1

ENDM

Pentru declararea datelor membre avem următoarele macrodefiniţii:

date MACRO den, tip, v


internDate %currId, den, tip, <v>

ENDM

internDate MACRO id, den, tip, v


No&id = No&id + 1


;alocare data


IFB <v>



den&id&@val@ tip ?


ELSE



den&id&@val@ tip v


ENDIF


;alocare protectie


IFE currProt



;; public



%OUT < define den as public ... >


ELSE



%OUT < define den as private ... >


ENDIF


den&id&Prot = currProt

ENDM

Deci se foloseşte macrodefiniţia date cu următoarea sintaxă:

date nume_data, tip_standard, [valoare]

Cu ajutorul acestei macroinstrucţiuni se alocă o variabilă de un anumit tip (precizat prin parametrul <tip>) care are un nume modificat (format prin stringizare sub şablonul <den><id_clasa>@val@) faţă de cel indicat de utilizator) cu scopul de restricţionare a accesului direct. Totodată se creşte numărul de elemente ale clasei şi se defineşte o variabilă (<den><id_clasa>Prot) prin care se va reţine protecţia asupra datei introduse.

Declararea metodelor membre se face în mod asemănător folosindu-se sintaxa:


func nume_rutina

iar macrodefiniţiile folosite sunt:

func MACRO den


internFunc %currId, den

ENDM

internFunc MACRO id, den


No&id = No&id + 1


IFE currProt



;; public



%OUT < define function den as public ... >


ELSE



%OUT < define function den as private ... >


ENDIF


den&id&FuncProt = currProt

ENDM

Aceste macrodefiniţii nu fac altceva decât să crească numărul de elemente ale clasei (No<id_clasa>) şi să stabilească protecţia pentru procedura respectivă (<den><id_clasa>FuncProt). În cadrul structurii nu este necesară reţinerea vreunei evidenţe a procedurii (pointer), rutina respectivă fiind identificată printr-un nume ce depinde de identificatorul obiectelor clasei respective. Pentru definirea funcţiei (după declararea claselor) se foloseşte următoarea sintaxă:


classFunc nume_clasa, nume_rutina


...


endFunc   nume_clasa, nume_rutina
ce are ca suport următoarele macrodefiniţii:

classFunc MACRO ob,fun


incepProc %(ob&Id), fun


;scoate temporar protectia pt datele din aceeasi clasa


setDisableProt %(ob&Id), 1

ENDM

incepProc MACRO id, fun


currApelator = id

;folosit in functiile prietene


currId = id


fun&id&@func@ proc

ENDM

endFunc MACRO ob,fun



;repune protectia



setDisableProt %(ob&Id), 0



;se iese din functie



currApelator = 0



currId = 0


ret


sfarProc %(ob&Id), fun

ENDM

sfarProc MACRO id, fun


fun&id&@func@ endp

ENDM
Se începe prin stabilirea unor variabile simbolice pentru identificarea obiectului în al cărui context se vor executa instrucţiunile din rutina (<currApelator>, folosit pentru accesul la datele claselor pentru care clasa curentă a fost declarată ca prietenă, şi <currId> pentru accesul la datele membre clasei). După care se declară începutul procedurii (se remarcă numele schimbat al procedurii: <denumirea_functiei><id_clasa>@func@) şi se întrerupe, pe parcursul întregii rutine, protecţia datelor membre (deoarece se rulează în contextul clasei) prin setarea pe 1 a variabilei disable<id_clasa>Prot. Sfârşitul procedurii, pe lângă încheierea efectivă a rutinei, repune protecţia şi resetează variabilele simbolice folosite la identificarea contextului.

Declararea unei clase ca prietene a alteia (cu acces complet asupra datelor membre) se realizează prin definirea unei variabile simbolice cu ajutorul macrodefiniţiei friend:

friend MACRO clase


IRP cl,clase



friend2 <cl>



%OUT < Class cl became friend of current class ... >


ENDM

ENDM

friend2 MACRO cl


friend3 %(cl&Id), %currId

ENDM

friend3 MACRO id, currId


friend&id&of&currId = 1
;defineste variabila 

;de prietenie

ENDM
Pentru moştenire simplă sau multiplă, s-au folosit următoarele macrodefiniţii (moştenirea este realizată doar pentru un singur nivel, dar poate fi extinsă prin folosirea de macrodefiniţii recursive):

publicExt MACRO b


IRP den,b



publicDer = 1


;mostenire publica



makeBaseName1 %currId, den



%OUT < Current class is derived from den ... >


ENDM

ENDM

privateExt MACRO b


IRP den,b



publicDer = 0


;mostenire private



makeBaseName1 %currId, den



%OUT < Current class is derived from den ... >


ENDM

ENDM

makeBaseName1 MACRO currId, den


baseNo&currId = baseNo&currId + 1


makeBaseId currId, %(baseNo&currId), %(den&Id)


makeBaseName2 %(baseNo&currId), currId, %(den&Id), den

ENDM

;macro-uri interne pentru extend

makeBaseId MACRO currId, currBaseNo, baseId
;creeaza ;variabila id a bazei


baseNo&currBaseNo&Of&currId&Id = baseId

ENDM

makeBaseName2 MACRO currBaseNo, currId, baseId, den




;aloca obiectul de baza


baseNo&currBaseNo&@base@ den <>


;calculeaza deplasamentul fata de clasa derivata


depl&baseId&In&currId = OFFSET baseNo&currBaseNo&@base@


;mostenire publica


IF publicDer



;declara clasa derivata ca prietena a clasei de baza



friend3 currId, baseId


ENDIF

ENDM

Se creşte mai întâi numărul de clase de bază ale clasei curente (baseNo<id_clasa>), apoi se trece la definirea unei variabile de identificare a clasei de bază (baseNo<nr_baza>Of<id_clasa> = <id_baza>) şi la declararea efectivă a structurii clasei de bază în cea derivată (cu numele schimbat baseNo<nr_baza>@base@) şi a deplasării acesteia în  cadrul clasei derivate (depl<id_baza>In<id_clasa>). Pentru accesul la toate datele clasei de bază (în cazul moştenirii publice) se face apel la un artificiu: se declară clasa derivată ca o clasă prietenă a clasei de bază.

După cum s-a arătat, datele membre ale unei clase au numele modificat şi deci necunoscut pentru utilizator. Pentru accesul la acestea, prin verificarea protecţiei şi căutarea în clasele de bază, s-au construit următoarele macrodefiniţii:

set MACRO ob, camp, v


;verifica daca suntem intr-o functie membra (nu se mai specifica


;






obiectul)


IFB <v>



operStanga movToOb, %currId, currOffset, ob, camp


ELSE



mov currOffSet,OFFSET ob



operStanga movToOb, %(ob&Id), currOffset, camp, v


ENDIF


IFE gasit

;nu s-a gasit



%OUT < **Error** camp is not a member of ob ... >



.ERR



ELSE




gasit = 0

;pt. urmatoarele cautari se revine ;la 0


ENDIF


deplasament = 0

ENDM

get MACRO unde, ob, camp


;verifica daca suntem intr-o functie membra (nu se mai ;specifica


;






obiectul)


IFB <camp>



operDreapta movFromOb, unde, %currId, currOffset, ob


ELSE



mov currOffSet,OFFSET ob



operDreapta movFromOb, unde, %(ob&Id), currOffSet, camp


ENDIF


IFE gasit

;nu s-a gasit



%OUT < **Error** camp is not a member of ob ... >



.ERR



ELSE




gasit = 0

;pt. urmatoarele cautari se revine ;la 0


ENDIF


deplasament = 0

ENDM

operStanga MACRO oper, id, ob, camp, v


IFDEF camp&id&@val@

;se afla direct in ;clasa?



;s-a gasit si se face operatia in functie de protectie



isProt? id, camp



oper id, ob, camp, v




gasit = 1
;l-a gasit


ELSE



;NU, poate se afla intr-o ;clasa de baza



;CAUTARE IN CLASELE DE BAZA:



searchInBase %(baseNo&id), id, camp



IF gasit




;se introduce operatia propriu-zisa




;baseId va contine id-ul la care s-a ajuns in ;cautare




;si a carui clasa cuprinde variabila




isProt? %baseId, camp
;vede daca este ;protejata








;in clasa derivata




oper %baseId, ob, camp, v








ENDIF


ENDIF

ENDM

operDreapta MACRO oper, unde, id, ob, camp


IFDEF camp&id&@val@

;se afla direct in ;clasa?



;s-a gasit si se face operatia in functie de protectie



isProt? id, camp



oper unde, id, ob, camp



gasit = 1
;l-a gasit


ELSE



;NU, poate se afla intr-o ;clasa de baza



;CAUTARE IN CLASELE DE BAZA:



searchInBase %(baseNo&id), id, camp



IF gasit




;se introduce operatia propriu-zisa




;baseId va contine id-ul la care s-a ajuns in ;cautare




;si a carui clasa cuprinde variabila




isProt? %baseId, camp
;vede daca este ;protejata








;in clasa derivata




oper unde, %baseId, ob, camp



ENDIF


ENDIF

ENDM

movToOb MACRO id, ob, camp, v


push bx


mov bx,word ptr ob


mov DS:[bx][deplasament].&camp&id&@val@, v


pop bx

endm

movFromOb MACRO unde, id, ob, camp


push bx


mov bx,word ptr ob


mov unde, DS:[BX][deplasament].&camp&id&@val@


pop bx

endm
Aceste macrodefiniţii, în cazul găsirii variabilelor, verifică protecţia acestora cu ajutorul macroinstrucţiunii:

isProt? MACRO id, den


isProt?2 id, den, %currApelator

ENDM

isProt?2 MACRO id, den, currApelator


;verifica daca suntem intr-o functie a unei clase ;prietene


IFNDEF friend&currApelator&of&id


;verifica daca se folosesc protectiile


IFE disable&id&Prot


IF den&id&Prot



%OUT < **Error** Member &den is not accesible ... >



.ERR


ENDIF


ENDIF


ENDIF

ENDM

În cazul negăsirii variabilei în clasa curentă se realizează căutarea ei în primul nivel de derivare cu ajutorul macrodefiniţiei:

searchInBase MACRO no, id, camp


noCurr = 1


REPT no


;cauta in toate clasele de baza



searchInBase2 %noCurr, id, camp



IF gasit


;l-a gasit si op. e deja ;introdusa




EXITM

;se iese din macrourile ;recursive



ENDIF



noCurr = noCurr + 1


ENDM

ENDM

searchInBase2 MACRO noCurr, currId, camp


baseId = baseNo&noCurr&Of&currId&Id


searchInBase3 currId, %baseId, camp

ENDM

searchInBase3 MACRO currId, baseId, camp


searchInBase4 camp&baseId&@val@


IF gasit

;la gasit si se calculeaza deplasamentul in




;functie de baseId



addDeplas %(depl&baseId&In&currId)


ENDIF

ENDM

searchInBase4 MACRO den


IFDEF den

;data cu denumirea den a fost ;definita



gasit = 1


ENDIF

ENDM

addDeplas MACRO deplCurr


deplasament = deplasament + deplCurr

ENDM 
Aceste macrodefiniţii, în cazul găsirii variabilei, setează <gasit> pe 1 şi încarcă <baseId> cu identificatorul clasei în care aceasta a fost găsită. Totodată se calculează şi deplasamentul clasei de bază în derivată (folosit de funcţiile de acces la membru).

Pentru apelul procedurilor membre mecanismul este acelaşi, chiar simplificându-se datorită necalculării deplasamentului (pentru apel s-au construit macrodefiniţii run, isFuncProt?, searchFuncInBase). De asemenea, pentru accesul la datele membre se pot imagina şi funcţii mai complicate, cum ar fi aflarea adresei sau alte operaţii directe fără preluarea conţinutului într-un registru.

Pentru exemplificarea utilizării acestor macrodefiniţii se va folosi exemplul numerelor reale şi complexe prezentat în prima variantă de implementare:

.model large 

include object.asm

.stack 100h

class Integer

public_


date
nr_int, dw, 0

;partea intreaga


func
Adunare


func
Inmultire

Integer ends

class Complex

publicExt <Integer>

public_


date
nr_complex, dw, 0
;partea complexa


func
Adunare


func
Inmultire

Complex ends

.code

;DEFINIREA METODELOR

classFunc Integer, Adunare


push ax


get ax, nr_int


add ax, word ptr [si]


set nr_int, ax


pop ax

endFunc   Integer, Adunare

classFunc Integer, Inmultire


push ax


get ax, nr_int


mul word ptr [si]


set nr_int, ax


pop ax

endFunc   Integer, Inmultire

classFunc Complex, Adunare


push ax


get ax, nr_int


add ax, word ptr [si]


set nr_int, ax


get ax, nr_complex


add ax, word ptr [di]


set nr_complex, ax


pop ax

endFunc   Complex, Adunare

classFunc Complex, Inmultire


push ax


push bx


get ax, nr_complex


mul word ptr [di]


mov bx, ax


get ax, nr_int


mul word ptr [si]


sub ax, bx


push ax


get ax, nr_complex


mul word ptr [si]


mov bx, ax


get ax, nr_int


mul word ptr [di]


add ax, bx




set nr_complex, ax


pop ax


set nr_int, ax


pop bx


pop ax

endFunc   Complex, Inmultire
Se observă modul de definire al claselor asemănător celui existent în limbajul C++. Clasa Complex este moştenită din clasa Integer (publicext <Integer>) ceea ce îi asigură şi partea întreagă. 

Numele procedurilor nu mai trebuie să fie diferit ca în exemplul anterior (deoarece numele procedurilor este format şi din identificatorul clasei), polimorfismul fiind realizat prin căutarea mai întâi în clasa curentă a procedurii şi apoi în cele de bază. În cadrul procedurilor accesul la datele membre se face prin funcţiile get şi set (obiectul nu mai trebuie precizat, se ia obiectul în contextul căruia are loc execuţia). Parametrii se trimit prin adresă prin registrele SI şi DI. Adresa obiectului nu mai trebuie trimisă ca în varianta anterioară, macrodefiniţiile folosindu-se de identificatorii obiectelor pentru a identifica structurile şi funcţiile corespunzătoare.

Pentru testare se poate folosi următoarea sursă:

.data


define ob_integer, Integer


define ob_complex, Complex


numar1
dw
10


numar2
dw
5

.code

start:


mov ax,@data


mov ds,ax


set ob_integer, nr_int, 4


set ob_complex, nr_int, 7


set ob_complex, nr_complex, 5


lea si, numar1


run ob_integer, Adunare


run ob_integer, Inmultire


lea di, numar2


run ob_complex, Adunare


run ob_complex, Inmultire

sfarsit:


mov ax,4c00h


int 21h

end start
Instanţierea claselor se face prin macrodefiniţia define. Valorile iniţiale ale datelor membre se pot seta prin instrucţiunea set (trebuie precizat obiectul de care aparţin datele membre). Înaintea apelului procedurilor de adunare şi înmulţire, se încarcă în registrele SI şi DI adresele numerelor folosite pentru aceste operaţii. Rezultatele acestor apeluri se regăsesc în noile valori ale datelor membre.

Pentru verificarea operaţiilor se procedează identic ca în cazul variantei anterioare (se foloseşte opţiunea /zi pentru asamblare şi opţiunea /v pentru link-editare după care se verifică valorile membrilor celor două obiecte prin execuţie pas cu pas în turbo debugger).

La compilarea acestui program se va obţine următorul listing:

E:\TOOLS\TASM\WORKING>tasm intcomp2

Turbo Assembler  Version 4.1  Copyright (c) 1988, 1996 Borland International

Assembling file:   intcomp2.ASM

< Creating class Integer ... >

< define nr_int as public ... >

< define function Adunare as public ... >

< define function Inmultire as public ... >

< Creating class Complex ... >

< Current class is derived from Integer ... >

< define nr_complex as public ... >

< define function Adunare as public ... >

< define function Inmultire as public ... >

Error messages:    None

Warning messages:  None

Passes:            1

Remaining memory:  422k
Macrodefiniţiile realizează o serie de afişări prin care indică structura claselor create şi eventualele erori.

Dacă se încearcă schimbarea definiţiei clasei Integer în următoarea (nr_int devine private):

class Integer

private_


date
nr_int, dw, 0

;partea intreaga

public_


func
Adunare


func
Inmultire

Integer ends

macrodefiniţiile vor semnala o serie de erori datorită încercării de iniţializare a părţii întregi prin instrucţiunile set (set ob_integer, nr_int, 4 şi set ob_complex, nr_int, 7) de după instanţierea obiectelor:

E:\TOOLS\TASM\WORKING>tasm intcomp2

Turbo Assembler  Version 4.1  Copyright (c) 1988, 1996 Borland International

Assembling file:   intcomp2.ASM

< Creating class Integer ... >

< define nr_int as private ... >

< define function Adunare as public ... >

< define function Inmultire as public ... >

< Creating class Complex ... >

< Current class is derived from Integer ... >

< define nr_complex as public ... >

< define function Adunare as public ... >

< define function Inmultire as public ... >

< **Error** Member nr_int is not accesible ... >

< **Error** Member nr_int is not accesible ... >

**Error** intcomp2.ASM(96) ISPROT?2(8) User error

**Error** intcomp2.ASM(97) ISPROT?2(8) User error

Error messages:    2

Warning messages:  None

Passes:            1

Remaining memory:  422k

Se observă că apar două erori, una pentru set-ul pe clasa Integer, iar alta pentru cel pentru clasa derivată Complex. Dacă cele două instrucţiuni set se şterg, procesul de compilare nu va mai semnala nici un fel de eroare, ceea ce demonstrează faptul că accesarea datelor membre private din cadrul procedurilor de adunare şi înmulţire sunt corecte datorită apartenenţei acestor rutine la cele două clase.

18.3.1. Optimizarea codului generat de macrodefiniţii

Aşa cum se va arăta spre sfârşitul capitolului, codul generat de macrodefiniţii în varianta actuală este ineficient. Spre exemplu, secvenţa de macroinstrucţiuni:


set ob_integer, nr_int, 4


lea si, numar1


run ob_integer, Adunare


run ob_integer, Inmultire
va genera următoarele instrucţiuni:



mov currOffSet,offset ob_integer



push bx



mov bx,word ptr currOffset



mov [bx][deplasament].nr_int1@val@, 4



pop bx



lea si, numar1



mov currOffset,offset ob_integer



call Adunare1@func@



mov currOffset,offset ob_integer



call Inmultire1@func@
După cum se observă, pentru indicarea instanţei curente (cea asupra căreia se realizează operaţiile) se foloseşte o dată de tip word, currOffset. Aceasta se încarcă în mod repetat în registrul BX, deoarece între instrucţiunile prezentate pot exista şi alte instrucţiuni care afectează acest registru. În consecinţă, codul rezultat din expandarea macroinstrucţiunilor este departe de a fi optim.

O optimizare foarte uşor de făcut, este aceea de a sacrifica un registru (aşa cum în cazurile anterioare s-a folosit registrul SI), care urmează să ia locul variabilei currOffset, dar care nu mai poate fi utilizat decât cu precauţie de către utilizator (trebuie salvat în stiva înainte de blocul de operaţii în care se foloseşte şi restaurat înainte de folosirea uneia dintre macroinstrucţiuni). Pentru aceasta s-a ales registrul BX, pentru că acesta oricum se foloseşte la adresarea bazată din cadrul macrodefiniţiilor set şi get.

După modificare, macrodefiniţiile get şi set devin:

;setare valoare

set MACRO ob, camp, v

;verifica daca suntem intr-o functie membra (nu se ;mai specifica obiectul)


IFB <v>



operStanga movToOb, %curId, curOffset, ob, camp


ELSE



mov bx,OFFSET ob



operStanga movToOb, %(ob&Id), ob, camp, v


ENDIF


IFE gasit

;nu s-a gasit


    %OUT < **Error** camp is not a member of ob ... >



.ERR



ELSE




gasit = 0

;pt. urmatoarele cautari se ;revine la 0


ENDIF


deplasament = 0

ENDM

;incarcare valoare

get MACRO unde, ob, camp


;verifica daca suntem intr-o functie membra (nu se ;mai specifica obiectul)


IFB <camp>

operDreapta movFromOb, unde, %curId, curOffset, ob


ELSE

mov bx,OFFSET ob

operDreapta movFromOb, unde, %(ob&Id), ob, camp


ENDIF


IFE gasit

;nu s-a gasit


    %OUT < **Error** camp is not a member of ob ... >



.ERR



ELSE




gasit = 0

;pt. urmatoarele cautari se ;revine la 0


ENDIF


deplasament = 0

ENDM

De asemenea, şi macroinstrucţiunile movToOb şi movFromOb folosite indirect (prin intermediul lui operDreapta şi operStanga) suferă modificări, nemaifiind necesară încărcarea în registrul bx a offset-ului:

movToOb MACRO id, camp, v


mov [BX][deplasament].&camp&id&@val@, v

endm

movFromOb MACRO unde, id, camp


mov unde, [BX][deplasament].&camp&id&@val@

endm
În mod asemănător se procedează şi pentru macrodefiniţiile folosite la apelul funcţiilor. Odată făcute aceste modificări, codul generat pentru secvenţa de instrucţiuni prezentată mai sus, se va micşora, producând o creştere semnificativă de viteză de execuţie:



mov bx,offset ob_integer



mov [bx][deplasament].nr_int1@val@, 4


lea si, numar1



mov bx,offset ob_integer



call Adunare1@func@



mov bx,offset ob_integer



call Inmultire1@func@
Optimizarea s-a produs pentru funcţia set prin preluarea directă a deplasamentului în registrul BX. O optimizare ce mai poate fi adusă codului, dar care ducă la o scădere a transparenţei implementării obiectelor, este încărcarea explicită în registrul BX a deplasamentului instanţei curente. Astfel, nu ar mai fi necesară încărcarea repetată a deplasamentului după cum se observă în codul următor:



mov bx,offset ob_integer



mov [bx][deplasament].nr_int1@val@, 4


lea si, numar1



call Adunare1@func@



call Inmultire@funct@
Făcându-se aceste simple modificări codul rezultat în urma expandării va fi asemănător cu cel oferit de prima variantă, cea care foloseşte structuri simple, dar va oferi în plus încapsulare, moştenire şi polimorfism, toate realizate în faza de asamblare. În acest mod, viteza de execuţie a programelor scrise în cele două variante va fi identică, dar avantajele oferite de cea de-a doua implementare vor fi net superioare.

18.4. Folosirea specificaţiilor proprii limbajului de asamblare

Începând cu versiunea 3.0, asamblorul produs de firma Borland conţine elementele fundamentale ce permit programarea orientată obiect în limbaj de asamblare, şi anume structurile de tip clasă. În Turbo Assembler, datele şi codul se pot încapsula în declaraţii speciale de tip STRUC, numite clase. O clasă (numită clasă derivată) poate moşteni codul şi datele unei alte clase (numită clasă de bază). Subrutinele clasei, sau metodele, pot fi statice (apelate direct) sau virtuale (apelate prin căutarea adresei subrutinei într-o tabelă a metodelor virtuale). 

Sintaxa unei declaraţii de clasă este:

<nume> STRUC <modificator> <nume_clasa> METHOD {



declaraţii metode

}



declaraţii date membre

ENDS <nume_clasa>

unde modificator poate fi NEAR sau FAR, însoţit sau nu de directiva GLOBAL.

Declaraţia metodelor clasei are următoarea sintaxă:

[virtual] nume_metoda : tip = nume_eticheta

Numele metodei nu este supus restricţiei de unicitate, ca în cazul declaraţiilor de membri ai tipului de date STRUC. Numele etichetei prin care este implementată metoda trebuie să fie unic. Tipul poate fi WORD sau DWORD.

Un exemplu de implementare a unei clase în limbaj de asamblare orientat obiect:

GLOBAL Baza_constructor: PROC

GLOBAL Baza_actiune: PROC

Baza STRUC GLOBAL METHOD{



constructor:WORD = Baza_constructor

VIRTUAL actiune:WORD = Baza_actiune

}

x
DD
0.0

ENDS Baza

Următoarele declaraţii definesc două obiecte ale clasei Baza, cu iniţializare sau fără:

DATASEG

b1
Baza
<1.1>

b2
Baza
< >
Prin folosirea moştenirii se creează clase noi din cele deja existente. Clasa derivată este o copie a clasei de bază la care se pot adăuga noi metode sau variabile. Metodele adăugate pot fi complet noi sau le pot înlocui pe cele cu acelaşi nume în clasa de bază. În plus, metodele din clasa nouă pot apela metodele din clasa de bază pe care le înlocuiesc. Nu se pot înlocui date ale clasei de bază.

În continuare se prezintă un exemplu de implementare a unei clase derivate din cea anterioară, în limbaj de asamblare orientat obiect:

GLOBAL Deriv_constructor: PROC

GLOBAL Deriv_actiune: PROC

Deriv STRUC GLOBAL Baza MEHOD{



constructor:WORD = Deriv_constructor

VIRTUAL actiune:WORD = Deriv_actiune

}

TBLPTR

y
DD
0.0

ENDS Deriv

Metodele virtuale diferă de metodele statice prin modul în care sunt adresate. În loc să calculeze adresa unei metode virtuale la asamblare, asamblorul generează instrucţiuni care extrag adresa la momentul execuţiei din VMT. Apelurile metodelor virtuale sunt indirecte, fiind făcute prin referinţă la intrările dintr-o VMT.

Inserarea uneia sau mai multor metode virtuale într-o declarare a unei clase se realizează prin prefaţarea declarării metodei cu cuvântul cheie VIRTUAL. Apoi, în secţiunea de date a clasei se inserează un pointer la VMT folosind directiva TBLPTR.

Cuvântul cheie TBLINST declară o instanţă a tabelei metodelor virtuale (VMT), definind spaţiul în memorie pentru aceasta. Simpla declarare a unei instanţe VMT nu realizează şi iniţializarea pointerului la VMT pentru diversele obiecte ale respectivei clase.

DATASEG

TBLINST

La următorul pas, pentru fiecare obiect în parte se include în constructorul său instrucţiunea TBLINIT <adresa>, unde adresa poate fi un registru sau o locaţie de memorie.

PROC
Deriv_constructor



TBLINIT Deriv PTR si


ret

ENDP Deriv_constructor

Pointerul la metoda statică constructor este iniţializat cu adresa subrutinei asociate, şi anume Deriv_constructor. Constructorul clasei trebuie apelat pentru toate obiectele clasei. Fiecare astfel de obiect are pointerul propriu la VMT, care trebuie iniţializat individual. Apelarea constructorului se realizează ca şi pentru oricare altă metoda statică. De exemplu, în segmentul de date, se defineşte mai întâi un obiect:

DATASEG

d1
Deriv

< >

În segmentul de cod, se adresează d1 prin intermediul ds:si şi se apelează constructorul clasei:

CODESEG

mov si, offset d1

CALL si METHOD Deriv:constructor

Apelarea metodei virtuale actiune din clasa Baza se exemplifică în continuare:

CODESEG

mov si, offfset b1

CALL Baza PTR si METHOD Baza:actiune

Dacă SI adresează un obiect al unei clase derivate, atunci subrutina actiune a clasei derivate este apelată. Astfel, în secvenţa de mai jos, deşi instrucţiunea CALL...METHOD specifică Baza, instrucţiunea de fapt apelează metoda actiune a clasei derivate.

CODESEG

mov si, offset d1

CALL Baza PTR si METHOD Baza:actiune

Aceasta este o dovadă a polimorfismului, proces prin care se creează clase care folosesc metode virtuale pentru a selecta acţiuni la momentul execuţiei. Deci obiectul însuşi determină care metodă virtuală să fie apelată.

În anumite cazuri, apelul de forma CALL...METHOD se poate înlocui cu JMP...METHOD pentru a optimiza metodele care se termină cu apeluri către alt metode.

Pentru a apela o funcţie virtuală a clasei de bază din interiorul unei funcţii virtuale a clasei derivate se foloseşte un apel de funcţie statică. De exemplu, în metoda virtuală actiune din clasa derivată Deriv se apelează metoda virtuală actiune din clasa de bază Baza:

PROC  Deriv_actiune



call
Baza_actiune


ret

ENDP Deriv_actiune
În continuare se reia exemplul numerelor întregi şi complexe pentru exemplificarea acestei variante.

Mai întâi este necesară folosirea unor directive:

IDEAL

JUMPS

LOCALS @@

MODEL large, PASCAL

STACK 1000h


. . . . .

Directiva IDEAL selectează modul de lucru Ideal al turbo asamblorului. Acest mod face ca membrii structurilor să fie locali acestora, ceea ce ne permite să avem nume de date membre identice în structuri diferite. Acest mod impune folosirea directivei GLOBAL pentru metode.

Cea de-a doua directivă, JUMPS, permite salturi condiţionale automate, făcând posibilă generarea unui cod mai eficient din partea asamblorului. Directiva LOCALS @@ previne o serie de conflicte rezultate din folosirea variabilelor cu acelaşi nume la nivel global şi la nivel de procedură. Prin MODEL large, PASCAL se declară modelul de memorie ca fiind large şi folosirea modului de transmitere al parametrilor ca în Pascal (parametrii sunt transmişi în ordinea în care sunt declaraţi), mai potrivit programării orientate obiect în limbajul de asamblare. STACK 1000h declară mărimea stivei.

Urmează codul în care se declară şablonul oferit de cele două clase şi procedurile din cadrul acestora:

...

GLOBAL Int_Adunare: PROC

GLOBAL Int_Inmultire: PROC

GLOBAL Complex_Adunare: PROC

GLOBAL Complex_Inmultire: PROC

STRUC Integer METHOD{


Adunare:WORD = Int_Adunare


Inmultire:WORD = Int_Inmultire

}

nr_int
DW
0

ENDS Integer

CODESEG

PROC Int_Adunare


ARG @@int:word


USES ax



mov ax, [@@int]


add ax, [(Integer PTR si).nr_int]


mov [(Integer PTR si).nr_int], ax


ret

ENDP Int_Adunare

PROC Int_Inmultire


ARG @@int:word


USES ax



mov ax, [@@int]


mul [(Integer PTR si).nr_int]


mov [(Integer PTR si).nr_int], ax


ret

ENDP Int_Inmultire

STRUC Complex Integer METHOD{


Adunare:WORD = Complex_Adunare


Inmultire:WORD = Complex_Inmultire

}

nr_complex DW
0

ENDS Complex

PROC Complex_Adunare


ARG @@int:word, @@compl:word


USES ax



CALL si METHOD Integer:Adunare, [@@int]


mov ax, [@@compl]


add ax, [(Complex PTR si).nr_complex]


mov [(Complex PTR si).nr_complex], ax


ret

ENDP Complex_Adunare

PROC Complex_Inmultire


ARG @@int:word, @@compl:word


USES ax, bx


mov ax, [@@compl]


mul [(Complex PTR si).nr_complex]


mov bx, ax


mov ax, [@@int]


mul [(Complex PTR si).nr_int]


sub ax, bx


push ax


mov ax, [@@compl]


mul [(Complex PTR si).nr_int]


mov bx, ax


mov ax, [@@int]


mul [(Complex PTR si).nr_complex]


add ax, bx


mov [(Complex PTR si).nr_complex], ax


pop ax


mov [(Complex PTR si).nr_int], ax


ret

ENDP Complex_Inmultire

...

Fiecare metodă a unei clase trebuie să lucreze cu o anumită instanţă a clasei respective. Această instanţă poate fi transmisă metodei prin oricare din modurile de transmitere a parametrilor, dar prin convenţie se transmite adresa acesteia prin perechea de registre DS:SI. De aceea accesul la o dată membră în cadrul unei metode a clasei se face prin (<nume_clasa> PTR si).<nume_camp>.

Parametrii sunt transmişi prin stivă, iar extragerea acestora se face prin directiva ARG, urmată de numele dat parametrului şi tipul acestuia. Folosirea prefixului @@ în numele parametrilor face ca aceştia să devină locali procedurii şi să nu fie confundaţi cu parametrii globali cu acelaşi nume. Se face uz de directiva USES prin care se indică regiştrii ce vor fi folosiţi, urmând ca asamblorul să includă push-urile şi pop-urile necesare pentru revenirea la valorile iniţiale ale acestor regiştri.

Pentru folosirea acestor clase se face apel la datele şi operaţiile de test prezentate şi la celelalte două variante:

...

DATASEG

ob_integer Integer <4>

ob_complex Complex <7, 5>

CODESEG

start:

mov ax,@data

mov ds,ax

mov si, offset ob_integer

CALL si METHOD Integer:Adunare, 10

CALL si METHOD Integer:Inmultire, 10

mov si, offset ob_complex

CALL si METHOD Complex:Adunare, 10,5

CALL si METHOD Complex:Inmultire, 10,5

sfarsit:

mov ax,4c00h

int 21h

end start
Înaintea apelurilor procedurilor se încarcă în registrul SI deplasamentul instanţei. Parametrii au fost transmişi prin specificarea unor constante după numele procedurii (numele este cel din cadrul clasei şi nu cel efectiv) în cadrul instrucţiunilor CALL. În locul acestor constante se puteau specifica şi identificatori de regiştri sau zone de memorie.

18.5. Analiza comparativă a variantelor de implementare a obiectelor

Fiecare dintre cele trei variante de implementare a obiectelor în limbajul de asamblare are atât avantaje cât şi dezavantaje. Prima variantă, cea prin care se foloseau structuri simple, este uşor de folosit şi nu necesită nici un fel de construcţii speciale. Prezintă însă o serie de deficiente: nu asigură o protecţie a datelor şi metodelor, în cazul moştenirii multiple apare problema deplasamentelor datelor membre ale claselor de bază, indicarea obiectelor se face explicit prin încărcarea într-un registru a adreselor, lipsesc elementele avansate de programare orientată obiect, cum este polimorfismul, clasele prietene şi funcţiile virtuale.

Cea de-a doua variantă, se remarcă printr-o transparenţă mare a mecanismelor de implementare a obiectelor, ceea ce poate uşura substanţial munca programatorului, precum şi prin optimalitatea codului datorită folosirii macrodefiniţiilor (durează mai mult compilarea, dar codul executabil rezultat este optim). Sintaxa a fost construită să semene cât mai mult cu cea folosită de limbajul C++, cu care majoritatea programatorilor sunt familiarizaţi. Varianta prezentată nu suportă moştenirea pe mai multe niveluri (dar care nu este imposibil de realizat datorită existenţei construcţiilor macro recursive şi repetitive) şi nici funcţiile virtuale, care necesită cod efectiv (nu numai macroinstrucţiuni) folosit în timpul executării programului. De asemenea, este necesară scrierea tuturor acestor macroinstrucţiuni şi includerea lor în fişierele sursă.

Ultima variantă, care foloseşte construcţii speciale orientate obiect specifice implementării propuse de firma Borland, acoperă multe din aspectele programării orientate obiect. În plus faţă de celelalte două variante, oferă folosirea funcţiilor virtuale. Trebuie ţinut cont că această implementare este specifică doar produsului realizat de firma Borland şi, deci, nu reprezintă un standard. Asambloarele oferite de alte firme nu cuprind aceste construcţii, iar dacă se doreşte un cod sursă compatibil (cu atât mai mult cu cât chiar sintaxa oferită de Borland poate suferi schimbări) trebuie făcut apel la alte variante. De asemenea, la fel ca în prima variantă, indicarea instanţelor nu se face la fel de uşor ca în cea de-a doua metodă, ci prin încărcarea explicită a adreselor în regiştri. Sintaxa nu permite o verificare completă a codului chiar înainte de link-editare, ca în cazul variantei folosirii macrodefiniţiilor, erorile fiind uneori greu de depistat, iar efectul acestora apare doar în timpul execuţiei.

Pentru o comparaţie mai profundă a celor trei variante de implementare, trebuie calculaţi şi o serie de indicatori, cum ar fi: numărul de instrucţiuni scrise, numărul efectiv de instrucţiuni folosite de asamblor, timpul de execuţie pentru o problemă dată etc.

Pentru a afla valorile unor astfel de indicatori în fiecare din variante trebuie apelat la un alt exemplu decât cel prezentat anterior. Astfel, pentru aflarea duratei de execuţie trebuie construit un exemplu de calcul care necesită un timp de execuţie mai lung, şi astfel cuantificabil. Se va lucra tot pe cele două clase prezentate în paginile de mai sus, dar pentru testarea operaţiilor acestora se va opta pentru un cod ciclic. Ca număr de iteraţii s-a optat pentru 10.000.000, ceea ce face ca timpul de execuţie să fie de ordinul zecilor de secunde. În cadrul unei iteraţii se realizează înmulţiri pentru ambele clase, utilizându-se datele de test deja prezentate.

Pentru prima variantă, cea care foloseşte doar structuri simple, noul cod va fi următorul:


; vechea porţiune de cod prin care se declara clasele


; si operaţiile acestora ...

...


contor1     dw
   2000


contor2     dw
   5000
 ; 2000 x 5000 = 10.000.000 de iteraţii

.code

start:


mov ax,@data


mov ds,ax



lea  si, numar1

;[si] = numar1, [si+2] = numar2



mov  cx, contor1

next1:



push cx


mov  cx, contor2

next2:



mov  ob_integer.nr_int, 4



mov  ob_complex.nr_int, 7



mov  ob_complex.nr_complex, 5



lea  di, ob_integer

 

call ob_integer.inm_int



lea  di, ob_complex



call ob_complex.inm_compl


loop next2



pop cx

loop next1

sfarsit:


mov ax,4c00h


int 21h

end start

Pentru varianta în care se folosesc macroinstrucţiuni, noul cod va fi:

; vechea porţiune de cod prin care se declara clasele

; si operaţiile acestora ...

...


contor1 dw
2000



contor2 dw
5000   ; 2000 x 5000 = 10.000.000 de iteraţii

.code

start:


mov ax,@data



mov ds,ax


lea  si, numar1


lea  di, numar2


mov  cx, contor1

next1:



push cx


mov  cx, contor2

next2:



set  ob_integer, nr_int, 4



set  ob_complex, nr_int, 7



set  ob_complex, nr_complex, 5

 

run  ob_integer, Inmultire



run  ob_complex, Inmultire



loop next2



pop cx

loop next1

sfarsit:


mov ax,4c00h


int 21h

end start

Iar pentru ultima variantă, în care se foloseşte implementarea Borland, codul folosit pentru testare va fi:

; vechea porţiune de cod prin care se declara clasele

; si operaţiile acestora ...

...


contor1    dw
2000



contor2    dw
5000   ; 2000 x 5000 = 10.000.000 ;de iteraţii

CODESEG

start:


mov ax,@data



mov ds,ax


mov  cx, contor1

next1:



push cx


mov  cx, contor2

next2:



mov si, offset ob_integer



mov [(Integer PTR si).nr_int], 4



CALL si METHOD Integer:Inmultire, [numar1]



mov si, offset ob_complex



mov [(Complex PTR si).nr_int], 7



mov [(Complex PTR si).nr_complex], 5



CALL si METHOD Complex:Inmultire, [numar1], [numar2]



loop next2



pop cx



loop next1

sfarsit:



mov ax,4c00h



int 21h

end start

Folosindu-se aceste secvenţe de cod se obţin indicatori atât cantitativi, cât şi calitativi referitori la cele trei variante implementate, tabelul 18.5.

Tabelul 18.1. Analiza comparativă a celor trei variante

Indicator
Varianta ce foloseşte structuri simple
Varianta ce foloseşte macroinstrucţiuni
Implementarea Borland orientată obiect



Varianta iniţială
Varianta optimizată


Linii de cod scrise
95
93
93
100

Linii de cod asamblate (după expandare)
95
147
96
121

Linii de cod folosite în timpul execuţiei
73
124
73
81

Timpul de execuţie (secunde)
20
37
22
27

Încapsulare (protecţia membrilor)

(
(


Moştenire

(
(
(

Polimorfism

(
(
(

Funcţii virtuale



(

18.5.1. Analiza indicatorilor rezultaţi

Tabelul prezintă patru indicatori, trei referindu-se la numărul liniilor de cod (ceea ce în limbaj de asamblare este echivalent cu numărul instrucţiunilor), iar ultimul este timpul de execuţie exprimat în secunde. Primul indicator se referă la numărul de linii scrise de programator (neincluzând codul sursă al macrodefiniţiilor), în timp ce al doilea reflectă numărul de instrucţiuni după efectuarea expandărilor macrodefiniţiilor. Al treilea indicator cuprinde doar instrucţiunile ce intervin în mod activ în timpul execuţiei (se elimină unele instrucţiuni cum ar fi etichetele, directivele de compilare etc.). Analiza se poate dezvolta apelându-se la lungimile în octeţi a instrucţiunilor. După aceşti patru indicatori se indică existenţa sau inexistenţa a patru concepte fundamentale ale programării orientate obiecte: încapsulare, moştenire, polimorfism şi funcţii virtuale.

Prima variantă de implementare prezintă cele mai puţine linii de cod şi cel mai mic timp de execuţie. Acest lucru este explicat prin lipsa celor mai elementare concepte ale programării orientate obiect (implementarea nu oferă nici măcar o protecţie a datelor sau polimorfism). Tot ceea ce scrie programatorul se reflectă în mod identic în codul sursă supus asamblării, singura abatere de la programarea tradiţională în cod de asamblare o constituie includerea pointerilor către proceduri în cadrul structurilor.

Cea de-a doua metodă de implementare este prezentată în cele două variante, observându-se câştigul substanţial de viteză al formei optimizate. Aceasta se apropie ca performanţă de metoda care foloseşte structuri simple (22 de secunde de execuţie faţă de 20), oferind în plus programatorului încapsulare, polimorfism şi moştenire multiplă. Prima variantă este mult mai lentă datorită folosirii unei date suplimentare pentru deplasare în cadrul structurilor şi a salvărilor repetate în stivă a registrului BX. Dar, aceasta oferă un grad mai ridicat de încapsulare. Utilizatorul macroinstrucţiunilor nu este supus nici unor restricţii în folosirea registrului BX, care în cazul variantei optimizate trebuie salvat şi restaurat de fiecare dată ţinând cont că şi codul expandat de macrodefiniţii foloseşte acest registru. De asemenea, varianta a doua, din motive de optimizare a codului, necesită încărcarea explicită în registrul BX a instanţei asupra căreia se lucrează, ceea ce duce la creşterea probabilităţilor de apariţie a erorilor.

Ultima variantă de implementare, cea oferită de firma Borland, oferă performanţe medii şi acest lucru se explică prin modul în care sunt trataţi parametrii procedurilor. Aceştia sunt transmişi prin stivă, iar pentru folosirea lor se rezervă regiştri care trebuiesc salvaţi înainte de încărcare şi restauraţi la ieşirea din procedură. Pentru a urmări codul generat, să luăm procedura de adunare a numerelor complexe:

PROC Complex_Adunare


ARG @@int:word, @@compl:word


USES ax



CALL si METHOD Integer:Adunare, [@@int]


mov ax, [@@compl]


add ax, [(Complex PTR si).nr_complex]


mov [(Complex PTR si).nr_complex], ax


ret

ENDP Complex_Adunare

Macrodefiniţiile ARG şi USES introduc următoarele modificări codului de mai sus:

PROC Complex_Adunare

PUSH    BP

MOV     BP,SP

PUSH    AX


CALL si METHOD Integer:Adunare, [@@int]


mov ax, [@@compl]


add ax, [(Complex PTR si).nr_complex]


mov [(Complex PTR si).nr_complex], ax

POP     AX

POP     BP


ret

ENDP Complex_Adunare

Pentru accesul la variabilele din stivă, trebuie salvată vechea valoare a registrului BP, care va fi folosit pentru accesarea stivei (mov BP, AX). Urmează salvarea registrului AX folosit pentru preluarea uneia dintre variabile. Înaintea revenirii din procedură se restaurează vechile valori ale regiştrilor AX şi BP. Pentru reutilizarea codului, este apelată metoda de adunare din clasa de bază, care, la rândul ei, va relua secvenţa de salvări şi restaurări de regiştri. Toate aceste operaţii duc la o încetinire semnificativă în execuţie atunci când apelurile de proceduri sunt folosite frecvent. Regiştrii, cum sunt AX şi BX, sunt salvaţi în mod repetat cu toate că păstrarea vechilor valori nu prezintă importanţă. Din acest punct de vedere, această metodă se aseamănă cu prima variantă a implementării prin macroinstrucţiuni, oferind un grad mare de transparenţă şi securitate, dar şi o încetinire a calculelor.

Alegerea uneia dintre variante trebuie făcută în strânsă corelaţie cu problema de rezolvat:

· dacă se dezvoltă cod relativ mic, care se doreşte a fi rapid se optează pentru prima variantă;

· dacă sunt necesare facilităţile de moştenire, încapsulare sau polimorfism pentru probleme mai complexe, dar pentru care timpul de execuţie este critic, se va opta pentru varianta optimizată ce foloseşte macroinstrucţiuni;

· dacă codul rezultat trebuie să fie sigur, punându-se accentul pe securitatea oferită în timpul execuţiei şi mai puţin pe timpul de execuţie, implementarea Borland orientată obiect sau prima variantă  ce foloseşte macrodefiniţii se dovedesc alegerile viabile;

· dacă se doresc secvenţe de program complexe, care apelează la lucrul cu pointeri spre obiecte ale căror tipuri sunt cunoscute abia în timpul execuţiei, cea mai buna alegere este implementarea Borland care oferă suport pentru funcţii virtuale.

Fiecare variantă este optimă pentru anumite tipuri de probleme şi cerinţe ale acestora, programatorului revenindu-i responsabilitatea alegerii. Prima variantă este simplă şi nu necesită nici un fel de efort suplimentar, dar lipsurile acesteia sunt evidente când se dezvoltă programe complexe. Variantele ce folosesc macrodefiniţii necesită scrierea sau procurarea acestora, iar implementarea oferită de firma Borland este recunoscută doar de asamblorul acesteia.

